Mathematica examples relevant to Gamma and Beta functions

Gamma function: Gamma[x]

Check that the defining integral indeed gives Gamma function

In[789]:=
```
Integrate[x^(p - 1) Exp[-x], {x, 0, Infinity}, Assumptions → Re[p] > 0]
```

Out[789]=
```
Gamma[p]
```

Check recursion relation (following quantity should equal 1)

In[795]:=
```
check[p_] = Gamma[p] p / Gamma[p + 1];
```

In[797]:=
```
Plot[check[p], {p, 0, 10}, PlotStyle → {Red, Thick}]
```

Out[797]=



Gamma[p] is indeed (p-1)! for integer p:

In[778]:=
```
{Gamma[7], 6!}
```

Out[778]=
```
{720, 720}
```

Plot shows the poles in the Gamma function on the real axis.

In[779]:=
```
Plot[Gamma[x], {x, -3, 3}, PlotStyle → Thick]
```

Out[779]=



Here’s a 3D plot of the absolute value of the Gamma function in the complex plane. Note that you can

rotate the view around.
Note the poles at x=0, -1, -2, -3,...

In[780]:=
```
Plot3D[Abs[Gamma[x + I y]], {x, -3, 3},
 {y, -3, 3}, PlotRange → {-1, 10}, AxesLabel → {"Re", "Im"}]
```

Out[780]=



This is the argument---a rather complicated plot!

In[781]:=
```
Plot3D[Arg[Gamma[x + I y]], {x, -3, 3}, {y, -3, 3}, AxesLabel → {"Re", "Im"}]
```

Out[781]=



Here follows the real and imaginary parts---a more complicated structure emerges around the poles.

In[782]:=
```
Plot3D[Re[Gamma[x + I y]], {x, -3, 3}, {y, -3, 3},
 PlotRange → {-10, 10}, AxesLabel → {"Re", "Im"}]
```

Out[782]=



In[783]:=
```
Plot3D[Im[Gamma[x + I y]], {x, -3, 3}, {y, -3, 3},
 PlotRange → {-10, 10}, AxesLabel → {"Re", "Im"}]
```

Out[783]=



For comparison a single pole

In[784]:=
```
Plot3D[Abs[1 / (x + I y)], {x, -3, 3}, {y, -3, 3},
 PlotRange → {-1, 10}, AxesLabel → {"Re", "Im"}]
```

Out[784]=



Here’s the real part , which is x/(x^2+y^2)

In[785]:=
```
Plot3D[Re[1 / (x + I y)], {x, -3, 3}, {y, -3, 3},
 PlotRange → {-10, 10}, AxesLabel → {"Re", "Im"}]
```

Out[785]=



...and the imaginary part which is -y/(x^2+y^2)

In[786]:=
```
Plot3D[Im[1 / (x + I y)], {x, -3, 3}, {y, -3, 3},
 PlotRange → {-10, 10}, AxesLabel → {"Re", "Im"}]
```

Out[786]=



Beta function: Beta[x,y]

The following integral defines Beta[x,y] for Re[p,q]>0
Mathematica jumps directly to the expression for Beta in terms of Gamma functions

In[798]:=
```
Integrate[x^(p - 1) (1 - x)^(q - 1), {x, 0, 1}]
```

Out[798]= ConditionalExpression$\left[\frac{\text{Gamma}[p]\ \text{Gamma}[q]}{\text{Gamma}[p+q]}, \text{Re}[q] > 0\ \&\&\ \text{Re}[p] > 0\right]$

Checking relation between Gamma and Beta functions

In[787]:=
```
{Beta[.5, .6], Gamma[.5] Gamma[.6] / Gamma[.5 + .6]}
```

Out[787]= {2.7745, 2.7745}

11.7 #2

In[799]:=
```
Integrate[Sqrt[Sin[x]^3 Cos[x]], {x, 0, Pi / 2}]
```

Out[799]= $\frac{\pi}{4\sqrt{2}}$